
1 Vorwort 

1.1 Idee 
Ich hatte eine SDIO-Karte, die ich im Standard aber nicht mit SD-Cards für mein 68008-System mit 
JADOS einsetzen konnte. Nun sollte die mal was tun – als universeller SPI-Controller … 

Und das Testopfer wurde der Soundrecorder-Baustein ISD17240. Idee war, einen Player analog zum 
NKC-CD-Player zu bauen. Aber leider kann man die Daten im FIFO-RAM des Bausteins nicht 
auslesen und schreiben. Sonst hätte man mit einer Festplatte oder SD-Card zusammen einen netten 
Player haben können. 

Also nur die SPI-Steuerung des Bausteins mit aufgenommenen und  intern fest gespeicherten Sounds 
… 

1.2 Ansatz 
Der Baustein ISD17240 soll via SPI mit der SDIO-Karte gesteuert werden. Deshalb an dieser Stelle 
einmal die Auffrischung der SPI-Basics (Quelle: Wikipedia): 

 

Eigenschaften  

- Die drei gemeinsamen Leitungen, an denen jeder Teilnehmer angeschlossen ist, sind: 

1. SCLK (englisch Serial Clock) auch SCK, wird vom Master zur Synchronisation 
ausgegeben 

2. MOSI oder SIMO (englisch Master Output, Slave Input) 
3. MISO oder SOMI (englisch Master Input, Slave Output) 

 
- Die Datenleitungen werden manchmal auch SDO (englisch Serial Data Out) und SDI 

(englisch Serial Data In) genannt, wobei die Benennung meistens aus der Sicht des 
jeweiligen Busteilnehmers erfolgt, so dass hier die Leitungen über Kreuz verbunden 
werden müssen. Statt SDI mit SDI und SDO mit SDO zu verbinden, müssen jeweils SDI 
mit SDO der Gegenstelle verbunden werden. 

 
- Eine oder mehrere mit logisch-0 aktive Chip-Select-Leitungen, welche alle vom Master 

gesteuert werden und von denen je eine Leitung pro Slave vorgesehen ist. Diese 
Leitungen werden je nach Anwendung unterschiedlich mit Bezeichnungen 
wie SS, CS, STE oder CE für Slave Select, Chip Select, Slave Transmit Enable oder Chip 
Enabler bezeichnet, oft noch in Kombination mit einer Indexnummer zur Unterscheidung.  

- voll duplexfähig 
 

- Viele Einstellmöglichkeiten, wie 
 

o mit welcher Taktflanke ausgegeben oder eingelesen wird 
o Wortlänge 
o Übertragung: MSB oder LSB zuerst 
o Unterschiedliche Taktfrequenzen bis in den MHz-Bereich sind zulässig. 

Viele Einstellungsmöglichkeiten sind unter anderem deshalb erforderlich, weil die 
Spezifikation für den SPI-Bus in vielen Eigenschaften nicht festgelegt ist, wodurch 
verschiedene, zueinander inkompatible Geräte existieren. Häufig ist beispielsweise für 
jeden angeschlossenen Schaltkreis eine eigene Konfiguration des steuernden 
Mikrocontrollers (Master des SPI-Bus) erforderlich. 

 

Protokollablauf und Einstellmöglichkeiten 

- An den Bus können so viele Teilnehmer angeschlossen werden, wie Slave-Select-Leitun-
gen vorhanden sind, zuzüglich des genau einen Masters, der seinerseits das Clock-
Signal an SCK erzeugt. Der Master legt mit der Leitung „Slave Select“ fest, mit welchem 
Slave er kommunizieren will. Wird sie gegen Masse gezogen, ist der jeweilige Slave aktiv 
und „lauscht“ an MOSI, bzw. legt er seine Daten im Takt von SCK an MISO. Es wird ein 
Wort vom Master zum Slave und ein anderes Wort vom Slave zum Master transportiert. 

- Ein Protokoll für die Datenübertragung wurde von Motorola zwar nicht festgelegt, doch 
haben sich in der Praxis vier verschiedene „Modi“ durchgesetzt. Diese werden durch die 
Parameter "Clock Polarity" (CPOL) und "Clock Phase" (CPHA) festgelegt. Bei CPOL=0 ist 
der Clock Idle Low, bei CPOL=1 ist der Clock Idle High. CPHA gibt nun an, bei der 
wievielten Flanke die Daten übernommen werden sollen. Bei CPHA=0 werden sie bei der 
ersten Flanke übernommen, nachdem SS auf Low gezogen wurde, bei CPHA=1 bei der 
zweiten. Somit werden die Daten bei CPOL=0 und CPHA=0 mit der ersten Flanke 
übernommen, die nur eine steigende Flanke sein kann. Bei CPHA=1 wäre es die zweite, 
also eine fallende Flanke. Bei CPOL=1 ist es folglich genau andersherum, bei CPHA=0 
fallende Flanke und bei CPHA=1 steigende Flanke. 

- Der Slave legt bei CPHA=0 seine Daten schon beim Runterziehen von SS an MISO an, 
damit der Master sie beim ersten Flankenwechsel übernehmen kann. Bei CPHA=1 
werden die Daten vom Slave erst beim ersten Flankenwechsel an MISO gelegt, damit sie 
beim zweiten Flankenwechsel vom Master übernommen werden können. Der Master 
hingegen legt seine Daten immer zum gleichen Zeitpunkt an, meist kurz nach der 
fallenden Flanke von SCK. 

- Mit jeder Taktperiode wird ein Bit übertragen. Beim üblichen Bytetransfer sind also acht 
Taktperioden für eine vollständige Übertragung nötig. Es können auch mehrere Worte 
hintereinander übertragen werden, wobei in der Spezifikation nicht festgelegt ist, ob 
zwischen jedem Wort das SS-Signal kurz wieder auf High gezogen werden muss. Eine 
Übertragung ist beendet, wenn das Slave-Select-Signal endgültig auf High gesetzt wird. 

- Es werden 4 SPI-Modes unterschieden 

Timings der 4 SPI-Modes (Quelle: introduction-to-spi-interface.pdf) 

1.3 Aktueller Stand 
Ich habe 2 Modifikationen an der SDIO-Karte vorgenommen, um sie universeller für SPI einsetzen zu 
können: 

- Einstellung der Taktpolarität (positiv / negativ) 
- Einstellung des IDLE-Pegels an MOSI (0 / 1) 

Des Weiteren habe ich eine Platine mit 2 Bausteinen ISD17240 erstellt, die sich an direkt an SV1 der 
SDIO-Karte anschließen lässt, sowie ein Programm in 6800x-Assembler, das die SDIO zusammen mit 
dieser Platine betreibt.  

 

2 Beschreibung des Konzepts  
Alles was man braucht, ist eine SDIO, an der 2 Modifikationen vorgenommen werden müssen, und ein 
per SPI anzusteuernder Baustein – im vorliegenden Fall die Platine mit 2 ISD17240. 

 

Modifikationen an der SDIO-Karte 

- Einstellung der Taktpolarität (positiv / negativ): 
o IC17 / 74HCT393 -  Pin 3 wird herausgebogen, sodass er nicht mehr in der IC-

Fassung sitzt. 
o Dieses Taktsignal wird auf einen neuen Jumper A1 geführt: 

 In Stellung „links“ wird es - wie vorher - an IC 09 / 74HCT00 - Pins 1,2 geführt 
 In Stellung „rechts“ wird es auf ein neues Single-Inverter-IC SN74LVC1G04 

geführt, bevor es an IC 09 / 74HCT00 - Pins 1,2 geführt wird 

 

- Einstellung des IDLE-Pegels an MOSI (0 / 1): 
o Der Pin 10 von IC 06 / 74HCT165 wird auf der Karte über den Widerstand R5 / 10 KΩ 

auf logisch 1 / +5 Volt gezogen. 
o Über einen weiteren, neuen Jumper A2 kann Pin 10 von IC 06 / 74HCT165 nun auf 

logisch 0 / 0 Volt gezogen werden. 

Folgender, per SPI zu steuernder Baustein wird eingesetzt   

 

Funktion Baustein Beschreibung Datenblatt 

Soundrecorder ISD17240 Soundrecorder mit FIFO-RAM ISD1700.pdf, 

ISD1700_Design_Guide.pdf, 

Design_Considerations_for_ISD1700_Fa

mily.pdf 

 

 

3 Einstellung der modifizierten SDIO-Karte für 
den Betrieb mit dem Baustein ISD17240 

 

3.1 SPI-Anforderungen des ISD17240 
 

 

 

Ergebnis 

- SPI-Mode 0  
- mit der Nebenbedingung, dass MOSI den IDLE-Wert logisch „0“ besitzen muss (siehe 2. 

Modifikation der SDIO-Karte) 
- Außerdem muss bei MOSI zuerst das LSB ausgegeben werden und bei MISO das LSB 

zuerst entgegen genommen werden. Das Drehen der Bytes wird in einem Unterprogramm 
des Testprogramms erledigt.  

 

 

 

 



 

NDR-Klein-Computer – SDIO als SPI-Controller mit Soundrecorder-IC   

Spezifikation Stand: 17.08.2020          Seite   9 von 22 

 

NDR-Klein-Computer – Digisound 

 s       nE
systematische

Prozesseneuer

Einführung

3.2 Einstellung der SDIO-Karte 
 

Die SDIO-Karte wird an den beiden neuen Jumpern folgendermaßen eingestellt: 

- Einstellung der Taktpolarität (positiv / negativ) – Jumper A1 auf „links“ 
- Einstellung des IDLE-Pegels an MOSI (0 / 1) – Jumper A2 gesteckt 

 

 

 

3.3 Der Baustein ISD17240 
 

Es handelt sich bei dem Baustein um einen Soundrecorder, der mit Tasten - stand alone - oder SPI 
bedient werden kann (Quelle: ISD1700_Design_Guide.pdf). 

 

Blockdiagramm 

Pin-Belegung 

Schaltplan und Testplatine 

SPI-Befehlssatz 

3.4 Testprogramm 
 

Das Testprogramm „NKCDS10.ASM“ besitzt folgende Funktionen: 

 

- Einstellung der SDIO-Karte: 
o Datenregister :   spidata equ $FFFFFF20*CPU 
o Statusregister:   spistat equ $FFFFFF21*CPU 

 Einstellung:     spiinit equ %11001011            

400 kHz, ReadOnly, Power on, kein 

                                       Geraet ausgewaehlt 

- Globale SPI-Einstellung: 
o Variable „spidev“ legt das SPI-Gerät fest 
o Variable „twistdat“ legt fest, ob ein zu übertragener (MOSI) und zu empfangender 

(MISO) Datenpuffer byteweise gedreht werden soll (MSB zu LSB) 
 

- Unterprogramme: 
o „Befehl“: 

 Ausführung eines in A0 übergebenen Befehls 
 Statusprüfung auf ERROR 

o „Status“: 
 Modus „init“ : 

 Wartet eine Zeit, Löscht INT und prüft auf RDY für nächsten Befehl 
 Modus „standard“: 

 Wartet auf INT, Löscht INT und prüft auf RDY für nächsten Befehl 
 

- Hauptprogramm: 
o Initialisierung SDIO-Karte 
o SPI-Gerät auswählen 
o Festlegung, dass Datenbytes gedreht werden 
o Power UP – Schalten in SPI-Modus 
o Audiopfad-Konfiguration 
o Play eines vorher per Taste aufgenommenen Sounds 
o Power Down 

;************************************************************************ 

;*                              NKC-Digisound                           * 

;************************************************************************ 

;* Steuerung des Soundrecorder-ICs ISD17240 via SPI - Bus               * 

;* Als SPI - Controller wird die NKC-SDIO-Karte verwendet.              * 

;* Basisadresse ist $FFFFFF20                                           * 

;************************************************************************ 

;* Testprogramm   Version 1.0   15.08.2020  by sEn                      * 

;*                                                                      * 

;* Funktionen:                                                          * 

;*      - Initialisierung SDIO-SPI-Controller                           * 

;*      - Initialisierung Baustein ISD17240                             * 

;*      - Ausführung von Befehlen an Baustein ISD17240                  * 

;*      - Statusueberwachung Baustein ISD17240                          * 

;************************************************************************ 

 

 

 start: 

 bra digsound 

 rts 

 

 

;*** Konstanten SDIO-SPI-Controller 

 

 cpu            equ 1                   ;68008 

 

 spidata        equ $FFFFFF20*CPU       ;SPI-Datenregister 

 spistat        equ $FFFFFF21*CPU       ;SPI-Statusregister 

 

 spiinit        equ %11001011           ;400 kHz, ReadOnly, Power on, kein 

                                        ;Geraet ausgewaehlt 

 spidev0        equ %11001010           ;Geraet 0 ausgewaehlt 

 spidev1        equ %11001001           ;Geraet 1 ausgewaehlt 

 

 ;*** Konstanten SPI-Commands fuer ISD 17240 

 

 pu:                    dc.b    $01,$00,$FF 

 stop:                  dc.b    $12,$00,$FF 

 reset:                 dc.b    $03,$00,$FF 

 clrint:                dc.b    $04,$00,$FF 

 rdstatus:              dc.b    $15,$00,$00,$FF 

 rdplaypt:              dc.b    $06,$00,$00,$00,$FF 

 pd:                    dc.b    $07,$00,$FF 

 rdrecpt:               dc.b    $08,$00,$00,$00,$FF 

 devid:                 dc.b    $09,$00,$00,$FF 

 play:                  dc.b    $50,$00,$FF 

 rec:                   dc.b    $51,$00,$FF 

 erase:                 dc.b    $52,$00,$FF 

 gerase:                dc.b    $53,$00,$FF 

 rdapc:                 dc.b    $44,$00,$00,$00,$FF 

 wrapc1:                dc.b    $45,$0B,$0D,$FF         ;eigene Konfiguration 

 wmrapc2:               dc.b    $65,$0B,$0D,$FF         ;eigene Konfiguration 

 wrnvcfg:               dc.b    $46,$00,$FF 

 ldnvcfg:               dc.b    $47,$00,$FF 

 fwd:                   dc.b    $48,$00,$FF 

 chkmem:                dc.b    $49,$00,$FF 

 extclk:                dc.b    $4A,$00,$FF 

 setplay:               dc.b    $80,$00,$00,$00,$00,$00,$00,$FF 

 setrec:                dc.b    $81,$00,$00,$00,$00,$00,$00,$FF 

 seterase:              dc.b    $82,$00,$00,$00,$00,$00,$00,$FF 

 

 ;*** Variablen 

 

  spidev:               df.b  1, $00            ;SPI-Geraet 

  twistdat:             df.b  1, $00            ;MOSI/MISO-Bytes drehen 

  miso:                 df.b 10, $FF            ;MISO-Puffer 

  datch:                df.b 10, $FF            ;MISO-Puffer gedreht 

 

 ds 0 

 

 

;*** Unterprogramme 

 

;Allgemein 

 

 twdat:                         ;Alle Bytes eines Datenpuffers werden gedreht - 

                                ;MSB zu LSB. A0 ist Eingagnspuffer. 

  lea datch(pc), A1             ;Zielpuffer der gedrehten Daten 

  combyte: 

   move.b (A0)+, d0             ;1 Byte der Originaldaten 

   cmp.b #$FF, d0               ;Endekennung eines Befehls 

   beq.s twdatend 

   clr.l d2 

   move.l #8, d3 

  chbyte:                       ;Sortiert Byte aus D0 mit MSB zu LSB in D1 

   subq #1, d3 

   btst d3, d0                  ;Abfrage Bitwert in D0 ... 

   beq.s null 

   bset d2, d1                  ;Wenn 1, dann in D1 setzen 

   bra.s chbytend 

   null: 

   bclr d2, d1                  ;Wenn 0, dann in D4 löschen 

   chbytend: 

   addq #1, d2 

   cmp.b #8, d2 

   bne.s chbyte 

   move.b d1, (A1)+ 

   bra.s combyte 

  twdatend: 

  move.b d0, (A1)               ;Endkennung des Befehls 

  lea datch(pc), A1             ;Puffer neu laden fuer Verwendung in Befehl 

 rts 

 

;SDIO-SPI-Controller 

 

 sdiobusy:                      ;Prueft bysy-Status des SDIO-SPI-Controllers 

  btst #0, spistat 

  bne.s sdiobusy 

 rts 

 

;ISD 1742-Baustein 

 

status:                         ;Prueft, ob Befehl fertig verarbeitet und ob 

                                ;bereit fuer naechsten Befehl 

                                ;D0=0 - monitor INT, CLR INT, check RDY 

                                ;D0=1 - delay, CLR INT, check RDY 

 cmp.b #1, d0                   ;Check auf init oder standard 

 beq.s puinit 

 lea rdstatus, A0               ;RDSTATUS-Befehl 

 bsr befehl                     ;und ausführen 

 move.b 0(A1), d0               ;1. MISO-Byte 

  btst.b #4, d0                 ;INT-Bit 

  beq.s status                  ;Wenn 1, dann aktuelle Operation fertig 

  bra.s initclr 

 puinit: 

   move #1, d0                  ;Warten auf power up 

   moveq #!delay, d7 

   trap #1 

 initclr: 

  lea clrint, A0                ;CLRINT-Befehl 

  bsr befehl                    ;und ausführen 

 chkrdy: 

  lea rdstatus, A0              ;RDSTATUS-Befehl 

  bsr befehl                    ;und ausführen 

  move.b 2(A1), d0              ;3. MISO-Byte 

  btst #0, d0                   ;RDY-Bit 

  beq.s chkrdy                  ;Wenn1, dann ready fuer naechsten Befehl 

 rts 

 

 befehl:                        ;Verarbeitet SPI-Befehle fuer ISD17240 

                                ;Der Befehl steht in A0, das SPI-Geraet in D0. 

                                ;D4=1, dan werden Daten gedreht MSB zu LSB. 

                                ;Wenn Fehler, dann erneute Ausfuehrung 

  movea A0, A1                  ;Wenn nicht gedreht wird, Befehl direkt 

  cmp.b #1, twistdat            ;Bytes für MOSI drehen? 

  bne.s spidevs 

  bsr twdat                     ;Der Befehl aus A0 steht gedreht in A1 

  spidevs: 

   cmp.b #1, spidev             ;Test auf SPI Geraet 1 

   bne.s dev0 

   move.b #spidev1, spistat     ;SPI-Geraet 1 mit CS1=0 auswählen 

   bra process 

   dev0: 

    move.b #spidev0, spistat    ;SPI-Geraet 1 mit CS0 = 0 

  process: 

  lea miso(pc), A2              ;MISO-Puffer laden 

  byte: 

   move.b (A1)+, d0             ;Byte aus Puffer holen und SPI-Transfer - MOSI 

   cmp.b #$FF, d0               ;bei $FF ist der Befehl zuende! 

   beq.s byteend 

   move.b d0, spidata           ;MOSI 

   bsr sdiobusy                 ;Warten bis SDIO-SPI-Controller ready ... 

   move.b spidata, (A2)+        ;und MISO 

   bra byte                     ;bis alle Bytes gesendet sind 

  byteend: 

   move.b d0, (A2)              ;MISO-Puffer-Abschluss 

   move.b spiinit, spistat      ;SPI-Geraete deaktivieren, Transaktion 

                                ;abschliessen 

  lea miso(pc), A1              ;Wenn nicht gedreht wird, MISO direkt 

  cmp.b #1, twistdat            ;Bytes für MISO drehen? 

  bne.s chkstat 

  lea miso(pc), A0 

  bsr twdat                     ;Die Daten aus A0 stehen gedreht in A1 

  chkstat:                      ;Statuspruefung 

  move.b 0(A1), d0              ;1. MISO-Byte 

  btst.b #0, d0                 ;Test auf CMDERR-Bit des Statusregisters 

  bne.s befehl                  ;Wenn 1, dann Befehlt erneut senden 

 rts 

 

 

;*** Hauptprogramm 

 

 digsound: 

  clr.l d0 

  clr.l d1 

  clr.l d2 

  clr.l d3 

   

 

 ;* Initialisierung SPI-Controller 

 

  move.b #spiinit, spistat              ;Einstellung SPI-Controller: 

                                        ;400 KHz,onlyrd=1,power=1,CS0=CS1=1 

  move.b #$00, spidata                  ;Shiftregister mit $00 laden fuer 

                                        ;MOSI default = 0 - Pullup-Widerstand 

                                        ;R5 auf SDIO auf 0V gejumpert! 

  move #1, d0                           ;Warten auf init 

  moveq #!delay, d7 

  trap #1 

 

 ;* SPI-Graet auswaehlen 

 

  ;clr.b spidev                          ;SPI-Geraet 0 

  move.b #1, spidev                     ;SPI-Geraet 1 

 

 ;* Festlegung, ob Datenbytes gedreht werden sollen - MSB zu LSB 

 

   ;clr.b twistdat                       ;n i c h t  drehen 

   move.b #1, twistdat                  ; drehen 

 ;* Initialisierung des Bausteins ISD17240 

 

  lea pu, A0                            ;Befehl "power up" - switch auf SPI 

  bsr befehl                            ;Befehl ausfuehren 

 

  move.b #1, d0                         ;Status - Init 

  bsr status                            ;Statuspruefung 

 

;* APC-Konfiguration - Audio-Einstellungen: 

 

  lea wmrapc2, A0                       ;APC-Schreibbefehl mit meiner Konfig. 

  bsr befehl 

  move #1, d0                           ;Status - Init 

  bsr status                            ;Statuspruefung 

 

;* Ausführung von Aktionen (record, play, stop, erase, setrecord, setplay ...) 

  lea play, A0                          ;PLAY-Befehl 

  bsr befehl                            ;Befehl ausfuehrenb 

 

  clr.b d0                              ;Status - Standard 

  bsr status 

 

  lea pd, A0                            ;Befehl Power Down 

  bsr befehl 

 

 rts 

 

end. 

4 Anhang 

4.1 Datenblätter TTL-Bausteine:  

4.1.1 SN74LVC1G04 
 

 

 

 

 

 

4.2 Verweis auf Datenblätter komplexer Bausteine und 
Spezifikationen / Quellennachweis 

 

Baustein/Objekt Funktion Datenblatt / Spezifikation 

SPI-Grundlagen SPI-Bus introduction-to-spi-interface.pdf 

IDS17240 Soundrecorder IC ISD1700.pdf, ISD1700_Design_Guide.pdf, 
Design_Considerations_for_ISD1700_Family.pdf 

 

 


